#-------------------------------------------------
#
# Project created by QtCreator 2011-03-13T12:21:03
#
#-------------------------------------------------

QT       += core

QT       -= gui
#QT       += gui

TARGET = scintillate
CONFIG   += console
CONFIG   -= app_bundle

TEMPLATE = app

HEADERS += DetectorConstruction.hh \
    PhysicsList.hh \
    PrimaryGeneratorAction.hh \
    CathodeHit.hh \
    CathodeSD.hh \
    Messenger.hh \
    RunAction.h \
    Scintillation.h

SOURCES += main.cpp \
    DetectorConstruction.cc \
    PhysicsList.cc \
    PrimaryGeneratorAction.cc \
    CathodeHit.cc \
    CathodeSD.cc \
    Messenger.cc \
    RunAction.cpp \
    Scintillation.cpp

LIBS += $$system(clhep-config --libs)
contains(QT,gui) {
  LIBS += $$system(geant4-config --libs)
  QMAKE_CXXFLAGS += $$system(geant4-config --cflags)
  DEFINES += G4UI_USE G4VIS_USE
  message(Compiling with GUI)
} else {
  LIBS += $$system(geant4-config --libs-without-gui)
  QMAKE_CXXFLAGS += $$system(geant4-config --cflags-without-gui)
  DEFINES += G4UI_USE_TCSH
  message(Compiling without GUI)
}

INCLUDEPATH += $$system(clhep-config --include |sed s/-I//)

DEFINES += REVISION=\"\\\"$$system(svnversion)\\\"\"
